

# SUITES et ÉQUATIONS DE RÉCURRENCE : SOLUTIONS DES EXERCICES

Bernard Dupont

[Bernard.Dupont@univ-lille1.fr](mailto:Bernard.Dupont@univ-lille1.fr)

## Exercice M1

### Enoncé

Résoudre les équations de récurrence suivantes, d'abord en donnant la solution générale puis en tenant compte de la condition initiale :

1.  $u(n) + 2u(n-1) = 3n^2 + 1$  et  $u(0) = 1$

2.  $u(n+1) = \frac{u(n)+1}{2}$  et  $u(0) = 3$

3.  $3u(n-1) - 3u(n) + n = 1$  et  $u(0) = 1$

4.  $3u(n) - u(n-1) = 3^n$  et  $u(0) = 0$

5.  $u(n) + 2u(n-1) + e^{-n} = 0$  et  $u(0) = \frac{2e}{2e+1}$

6.  $u(n) - \frac{1}{2}u(n-1) = \frac{1}{2^n} - 1$  et  $u(0) = 0$

7.  $2u(n) - u(n-1) = n^2 + \frac{1}{2^{n-1}}$  et  $u(0) = 1$

### Solution

Dans tous les cas, il s'agit d'équations de récurrence linéaires du premier ordre à coefficients constants et avec second membre du type constant, polynômial, exponentiel ou mixte. La commande **rsolve** résout toujours ce genre de problème.

1.

```
> eqrec1:=u(n)+2*u(n-1)=3*n^2+1;#second membre polynômial
init1:=u(0)=1;
soll:=rsolve(eqrec1,u(t));#requête d'une solution générale
soll1:=rsolve({eqrec1,init1},u);#requête d'une solution
particulière
```

$$eqrec1 := u(n) + 2u(n-1) = 3n^2 + 1$$

$$init1 := u(0) = 1$$

$$soll := u(0) (-2)^t + 2(t+1) \left( \frac{1}{2}t + 1 \right) - \frac{5}{3}t - \frac{13}{9} - \frac{5}{9}(-2)^t$$

$$soll1 := \frac{4}{9}(-2)^n + 2(n+1) \left( \frac{1}{2}n + 1 \right) - \frac{5}{3}n - \frac{13}{9}$$

2.

```
> eqrec2:=u(n+1)=1/2*(u(n)+1);#second membre "déguisé"
```

```

init2:=u(0)=3;
sol2:=rsolve(eqrec2,u(t));#requête d'une solution générale
sol22:=rsolve({eqrec2,init2},u);#requête d'une solution
particulière

```

$$eqrec2 := u(n+1) = \frac{1}{2} u(n) + \frac{1}{2}$$

$$init2 := u(0) = 3$$

$$sol2 := u(0) \left(\frac{1}{2}\right)^t - \left(\frac{1}{2}\right)^t + 1$$

$$sol22 := 2 \left(\frac{1}{2}\right)^n + 1$$

3.

```

> eqrec3:=3*u(n-1)-3*u(n)+n=1;#équation écrite sous forme non
standard

```

```

init3:=u(0)=1;

```

```

sol3:=rsolve(eqrec3,u(t));#requête d'une solution générale
sol33:=rsolve({eqrec3,init3},u);#requête d'une solution
particulière

```

$$eqrec3 := 3 u(n-1) - 3 u(n) + n = 1$$

$$init3 := u(0) = 1$$

$$sol3 := u(0) - \frac{1}{3} - \frac{2}{3} t + \frac{1}{3} (t+1) \left(\frac{1}{2} t + 1\right)$$

$$sol33 := \frac{2}{3} - \frac{2}{3} n + \frac{1}{3} (n+1) \left(\frac{1}{2} n + 1\right)$$

4.

```

> eqrec4:=3*u(n)-u(n-1)=3^n;#second membre exponentiel de base
3

```

```

init4:=u(0)=0;

```

```

sol4:=rsolve(eqrec4,u(t));#requête d'une solution générale
sol44:=rsolve({eqrec4,init4},u);#requête d'une solution
particulière

```

$$eqrec4 := 3 u(n) - u(n-1) = 3^n$$

$$init4 := u(0) = 0$$

$$sol4 := u(0) \left(\frac{1}{3}\right)^t + \frac{3}{8} 3^t - \frac{3}{8} \left(\frac{1}{3}\right)^t$$

$$sol44 := \frac{3}{8} 3^n - \frac{3}{8} \left(\frac{1}{3}\right)^n$$

5.

```

> eqrec5:=u(n)+2*u(n-1)+exp(-n);#second membre exponentiel de
base e

```

```

init5:=u(0)=2*exp(1)/(2*exp(1)+1);

```

```

sol5:=rsolve(eqrec5,u(t));#requête d'une solution générale

```

`sol55:=rsolve({eqrec5,init5},u);#requête d'une solution particulière`

$$eqrec5 := u(n) + 2 u(n - 1) + e^{-n}$$

$$init5 := u(0) = \frac{2e}{2e + 1}$$

$$sol5 := u(0) (-2)^t - \frac{\left(\frac{1}{e}\right)^t}{2e + 1} + \frac{(-2)^t}{2e + 1}$$

$$sol55 := \frac{2e(-2)^n}{2e + 1} - \frac{\left(\frac{1}{e}\right)^n}{2e + 1} + \frac{(-2)^n}{2e + 1}$$

6.

`> eqrec6:=u(n)-1/2*u(n-1)=(1/2^n)-1;#second membre "mixte"  
init6:=u(0)=0;`

`sol6:=rsolve(eqrec6,u(t));#requête d'une solution générale  
sol66:=rsolve({eqrec6,init6},u);simplify(%)#requête d'une solution particulière`

$$eqrec6 := u(n) - \frac{1}{2} u(n - 1) = \frac{1}{2^n} - 1$$

$$init6 := u(0) = 0$$

$$sol6 := u(0) \left(\frac{1}{2}\right)^t - 2 + \left(\frac{1}{2}\right)^t + \frac{1}{4} (4t + 4) \left(\frac{1}{2}\right)^t$$

$$sol66 := -2 + \left(\frac{1}{2}\right)^n + \frac{1}{4} (4n + 4) \left(\frac{1}{2}\right)^n$$

$$-2 + 2^{-n+1} + 2^{-n} n$$

7.

`> eqrec7:=2*u(n)-u(n-1)=n^2+1/2^(n-1);#second membre mixte  
init7:=u(0)=1;`

`sol7:=rsolve(eqrec7,u(t));#requête d'une solution générale  
sol77:=rsolve({eqrec7,init7},u);#requête d'une solution particulière`

$$eqrec7 := 2 u(n) - u(n - 1) = n^2 + \frac{1}{2^{n-1}}$$

$$init7 := u(0) = 1$$

$$sol7 := u(0) \left(\frac{1}{2}\right)^t + \frac{1}{4} (4t + 4) \left(\frac{1}{2}\right)^t - 5t + 1 - 4 \left(\frac{1}{2}\right)^t + 2(t + 1) \left(\frac{1}{2} t + 1\right)$$

$$sol77 := -3 \left(\frac{1}{2}\right)^n + \frac{1}{4} (4n + 4) \left(\frac{1}{2}\right)^n - 5n + 1 + 2(n + 1) \left(\frac{1}{2} n + 1\right)$$

## Exercice M2

### Enoncé

Résoudre les équations de récurrence suivantes, d'abord en donnant la solution générale puis en tenant compte des conditions initiales :

1.  $u(n) - 5u(n-1) + 6u(n-2) = 2n + 1$  avec  $u(0) = 1$  et  $u(1) = 4$

2.  $u(n+2) - 2u(n+1) + 2u(n) = 2^n$  avec  $u(0) = 0$  et  $u(1) = 1$

3.  $2\sqrt{3}u(n-1) - 4u(n-2) + 2^n \cos\left(\frac{1}{6}n\pi\right) = 0$  avec  $u(0) = 0$  et  $u(1) = 2 + \sqrt{3}$

### **Solution**

Il s'agit d'équations de récurrence linéaires du second ordre à coefficients constants et avec second membre du type polynômial, exponentiel ou mixte. La commande **rsolve** résout toujours ce genre de problème.

1.

```
> eqrec1:=u(n)-5*u(n-1)+6*u(n-2)=2*n+1;#second membre
polynomial de degré 1
init1:=u(0)=1,u(1)=4;
sol1:=rsolve(eqrec1,u(n));#recherche de la solution générale
sol11:=rsolve({eqrec1,init1},u);#recherche de la solution
avec conditions initiales
```

$$eqrec1 := u(n) - 5u(n-1) + 6u(n-2) = 2n + 1$$

$$init1 := u(0) = 1, u(1) = 4$$

$$sol1 := -(2u(0) - u(1))3^n - (-3u(0) + u(1))2^n + 33^n + 4 + n - 72^n$$

$$sol11 := 53^n - 82^n + 4 + n$$

2.

```
> eqrec2:=u(n+2)-2*u(n+1)+2*u(n)=2^n;#second membre exponentiel
de base 2
init2:=u(0)=0,u(1)=1;
sol2:=rsolve(eqrec2,u(n));#recherche de la solution générale
sol22:=rsolve({eqrec2,init2},u);#recherche de la solution
avec conditions initiales
```

$$eqrec2 := u(n+2) - 2u(n+1) + 2u(n) = 2^n$$

$$init2 := u(0) = 0, u(1) = 1$$

$$sol2 := \left(-\frac{1}{4} - \frac{1}{4}I\right) (-u(1) - Iu(1) + 2Iu(0)) (1 - I)^n + \left(\frac{1}{4} - \frac{1}{4}I\right) (u(1)$$

$$- Iu(1) + 2Iu(0)) (1 + I)^n + \left(-\frac{1}{4} - \frac{1}{4}I\right) (1 - I)^n + \left(-\frac{1}{4} + \frac{1}{4}I\right) (1 + I)^n$$

$$+ \frac{1}{2} 2^n$$

$$sol22 := \left(-\frac{1}{4} - \frac{1}{4}I\right) (1 + I)^n + \left(-\frac{1}{4} + \frac{1}{4}I\right) (1 - I)^n + \frac{1}{2} 2^n$$

3.

```
> eqrec3:=u(n)=2*sqrt(3)*u(n-1)-4*u(n-2)+2^n*cos(n*Pi/6);
#second membre mixte
init3:=u(0)=0,u(1)=2+sqrt(3);
sol3:=rsolve(eqrec3,u(n));#recherche de la solution générale
```

**sol33:=rsolve({eqrec3,init3},u);#recherche de la solution avec conditions initiales**

$$eqrec3 := u(n) = 2\sqrt{3} u(n-1) - 4u(n-2) + 2^n \cos\left(\frac{1}{6} n \pi\right)$$

$$init3 := u(0) = 0, u(1) = 2 + \sqrt{3}$$

$$sol3 := \frac{1}{2} I(-\sqrt{3} u(0) - Iu(0) + u(1)) (\sqrt{3} - I)^n + \frac{1}{2} I(\sqrt{3} u(0) - Iu(0)$$

$$- u(1)) (\sqrt{3} + I)^n - 2 \cdot 2^n \cos\left(\frac{1}{6} n \pi\right) - 96 \sqrt{3} 2^{n-5} \cos\left(\frac{1}{6} (n-5) \pi\right)$$

$$- 24 \sqrt{3} 2^{n-3} \cos\left(\frac{1}{6} (n-3) \pi\right) + 144 2^{n-4} \cos\left(\frac{1}{6} (n-4) \pi\right)$$

$$+ 6 \sqrt{3} 2^{n-1} \cos\left(\frac{1}{6} (n-1) \pi\right) - 12 2^{n-2} \cos\left(\frac{1}{6} (n-2) \pi\right)$$

$$+ 64 2^{n-6} \cos\left(\frac{1}{6} (n-6) \pi\right)$$

$$sol33 := -\frac{1}{2} I(2 + \sqrt{3}) (\sqrt{3} + I)^n - \frac{1}{2} I(-2 - \sqrt{3}) (\sqrt{3} - I)^n - 2 \cdot 2^n \cos\left(\frac{1}{6} n \pi\right)$$

$$- 96 \sqrt{3} 2^{n-5} \cos\left(\frac{1}{6} (n-5) \pi\right) - 24 \sqrt{3} 2^{n-3} \cos\left(\frac{1}{6} (n-3) \pi\right)$$

$$+ 144 2^{n-4} \cos\left(\frac{1}{6} (n-4) \pi\right) + 6 \sqrt{3} 2^{n-1} \cos\left(\frac{1}{6} (n-1) \pi\right)$$

$$- 12 2^{n-2} \cos\left(\frac{1}{6} (n-2) \pi\right) + 64 2^{n-6} \cos\left(\frac{1}{6} (n-6) \pi\right)$$

### Exercice M3

#### Enoncé

On reprend l'équation de récurrence qui n'a pas pu être résolue avec **rsolve**, soit

$$u_n = \frac{u_{n-1}^2 + 3}{2u_{n-1} + 2} \text{ avec } u_0 = 2. \text{ Montrer graphiquement que la suite } (u) \text{ est convergente.}$$


#### Solution

On commence par visualiser le diagramme en escalier avec la méthode graphique exposée dans le cours. Rappelons qu'il s'agit de visualiser une dynamique discrète par un diagramme en escalier.


```
> restart;
f:=x-(x^2+3)/(2*x+2);#définition de la fonction
u:=n->f(u(n-1));#définition de la récurrence
u(0):=2;#assignation du premier terme
graf1:=plot([f(x),x],x=0..3,color=[black,black],thickness=2)
:#graphe de la fonction f
marche:=i->([u(i),u(i)], [u(i),u(i+1)]):#fonction-procédure
marche
escalier:=k->seq(marche(i),i=1..k):#fonction-procédure
escalier à partir de la première marche
```

```
graf2:=plot([[u(0),0],[u(0),u(1)],escalier(3)]):#tracé de  
l'escalier à partir de la condition initiale  
with(plots):display({graf1,graf2});
```

$$f := x \rightarrow \frac{x^2 + 3}{2x + 2}$$
$$u := n \rightarrow f(u(n-1))$$
$$u(0) := 2$$



La présentation du graphique est améliorée en utilisant l'option **Text** de **Drawing** dans la barre contextuelle. On obtient alors :



On constate bien que la suite converge vers un équilibre dont il est facile de déterminer les coordonnées en résolvant l'équation  $l = \frac{l^2 + 3}{2l + 2}$ .

```
> solve(f(1)=1,1);
-3, 1
```

Il y a deux solutions mais on retient  $l = 1$  au vu du graphique.

## Exercice M4

### Enoncé

Soit la suite  $u_n = u_{n-1}^\alpha$  pour  $\alpha$  réel strictement positif différent de 1. On sait que si  $\alpha < 1$ , alors la suite est convergente et de limite 1 et si  $\alpha > 1$ , alors la suite diverge. Visualisez le comportement de cette suite dans les cas  $\alpha = 0.25$  et  $\alpha = 1.5$ .

### Solution

On commence par créer une fonction de deux variables  $f(\alpha, x) = x^\alpha$  qui sera particularisée en  $f1(0.25, x) = x^{0.25}$  et  $f2(1.5, x) = x^{1.5}$  pour traiter les deux cas posés dans la question.

```
> restart; f := (alpha, x) -> x^alpha; f1 := x -> f(0.25, x); f2 := x -> f(1.5, x); #préparation
```

```
f := (alpha, x) -> x^alpha
f1 := x -> f(0.25, x)
f2 := x -> f(1.5, x)
```


Reste à étudier séparément les deux cas, en appliquant la méthode "escalier". On choisit une condition initiale à sa convenance, dans ce qui suit  $u_0 = 0.25$  quand  $\alpha = 0.25$  et  $u_0 = 0.75$  quand

$\alpha = 1.5$ . L'amélioration des graphiques est effectuée sans faire appel à l'option Drawing du menu contextuel; elle est programmée directement dans les options de **display**.

Cas  $\alpha = 0.25$  :

```
> u:=n->f1(u(n-1)):
u(0):=0.25:
graf11:=plot([f1(x),x],x=0..1.7,y=0..1.7,color=[black,black],
thickness=2):
marche1:=i->([u(i),u(i)],[u(i),u(i+1)]):
escalier1:=k->seq(marche1(i),i=1..k):
graf12:=plot([[u(0),0],[u(0),u(1)],escalier1(3)]):
with(plots):
graf13:=display({graf11,graf12},textplot({[1.64,1.04,typeset
(y=f1(x))],
[1.5,1.65,typeset(y=x)]}),font=[TIMES,14],title=typeset("Cas
",alpha <1),titlefont=[TIMES,BOLD,16],xtickmarks=[0,0.25,0.5,
1,1.5],ytickmarks=[0.5,1,1.5]):graf13;
```

### Cas $\alpha < 1$



Cas  $\alpha = 1.5$  :

```
> v:=n->f2(v(n-1)):
v(0):=0.75:
graf21:=plot([f2(x),x],x=0..1.7,y=0..1.7,color=[black,black],
thickness=2):
marche2:=i->([v(i),v(i)],[v(i),v(i+1)]):
escalier2:=k->seq(marche2(i),i=1..k):
```




```

graf22:=plot([[v(0),0],[v(0),v(1)],escalier2(7)]):
graf23:=display({graf21,graf22},textplot({[1.2,1.6,typeset(y=
f2(x))],
[1.55,1.4,typeset(y=x)]}),font=[TIMES,14],title=typeset("Cas
",alpha >1),titlefont=[TIMES,BOLD,16],xtickmarks=[0,0.5,0.75,
1,1.5],ytickmarks=[0.5,1,1.5]):graf23;

```

### Cas $1 < \alpha$



## Exercice E1 (Multiplicateur-accélérateur de Hicks-Samuelson)

### Enoncé

Dans sa version la plus simple, le modèle du multiplicateur-accélérateur de Hicks-Samuelson a pour forme structurelle :

(1)  $Z_t = Y_t$  (équilibre sur le marché des biens et services)

(2)  $Z_t = C_t + I_t$  (la demande globale est la somme de la consommation des ménages et de l'investissement des firmes. Il n'y a pas d'Etat et pas de relations avec le reste du monde)

(3)  $C_t = c Y_{t-1}$  (fonction de consommation keynésienne avec retard)

(4)  $I_t = v (Y_{t-1} - Y_{t-2})$  (fonction d'investissement dans l'optique théorique d'un effet d'accélération)

où  $Y_t$  l'offre globale en  $t$ ,  $C_t$  la consommation des ménages,  $I_t$  l'investissement des firmes. La propension à consommer  $c$  est comprise entre 0 et 1. Le coefficient d'accélération  $v$  est strictement positif.

1. Montrer que l'évolution de l'offre est régie par une équation de récurrence linéaire d'ordre 2.

2. Résoudre cette équation de récurrence dans le cas général à l'aide de **rsolve**. Quels commentaires vous inspire l'output?

3. Mener une étude classique de l'équation de récurrence à partir de son équation caractéristique.

4. Représenter graphiquement les différents types de trajectoires-solutions suivant les valeurs prises

par  $c$  et  $v$ .

### Solution

1. On commence par écrire les 4 équations qui définissent la structure du modèle. Il vaut mieux éviter d'indicer les variables comme on le ferait à la main car il arrive que cette écriture soit mal interprétée par Maple.

```
> restart;
eq1:=Z(t)=Y(t);
eq2:=Z(t)=C(t)+Iv(t);
eq3:=C(t)=c*Y(t-1);
eq4:=Iv(t)=v*(Y(t-1)-Y(t-2));
      eq1 := Z(t) = Y(t)
      eq2 := Z(t) = C(t) + Iv(t)
      eq3 := C(t) = c Y(t - 1)
      eq4 := Iv(t) = v (Y(t - 1) - Y(t - 2))
```

(5.1)

Le repérage de l'équation de récurrence sur  $Y$  se fait en utilisant d'abord **solve** sur toutes les variables endogènes, puis par extraction de la solution en  $Y$ .

```
> solg:=solve({eq1,eq2,eq3,eq4},{Y(t),C(t),Iv(t),Z(t)});
solY:=subs(solg,Y(t));
eqrY:=Y(t)=solY;#équation de récurrence
solg := {C(t) = c Y(t - 1), Iv(t) = v Y(t - 1) - v Y(t - 2), Y(t) = c Y(t - 1) + v Y(t - 1)
      - v Y(t - 2), Z(t) = c Y(t - 1) + v Y(t - 1) - v Y(t - 2)}
      solY := c Y(t - 1) + v Y(t - 1) - v Y(t - 2)
      eqrY := Y(t) = c Y(t - 1) + v Y(t - 1) - v Y(t - 2)
```

(5.2)

On est bien en présence d'une équation de récurrence linéaire d'ordre 2, à coefficients constants et sans second membre.

2. Maple résout toutes les équations de récurrence linéaires, a fortiori **eqrY**.

```
> Ysol:=rsolve(eqrY,Y(t));
```

$$Ysol := \left( (-2 v Y(0) c - v^2 Y(0) - v Y(0) \sqrt{c^2 + 2 c v + v^2 - 4 v} - c^2 Y(0) \right. \\ \left. - c Y(0) \sqrt{c^2 + 2 c v + v^2 - 4 v} + 2 v Y(0) + Y(1) c + Y(1) v \right. \\ \left. + Y(1) \sqrt{c^2 + 2 c v + v^2 - 4 v} \right) \left( -\frac{2 v}{-c - v - \sqrt{c^2 + 2 c v + v^2 - 4 v}} \right)^t / \\ \left( \sqrt{c^2 + 2 c v + v^2 - 4 v} (-c - v - \sqrt{c^2 + 2 c v + v^2 - 4 v}) \right) + \left( (2 v Y(0) c \right. \\ \left. + v^2 Y(0) - v Y(0) \sqrt{c^2 + 2 c v + v^2 - 4 v} + c^2 Y(0) \right. \\ \left. - c Y(0) \sqrt{c^2 + 2 c v + v^2 - 4 v} - 2 v Y(0) - Y(1) c - Y(1) v \right)$$

(5.3)

$$\frac{+Y(1)\sqrt{c^2+2cv+v^2-4v}\left(-\frac{2v}{-c-v+\sqrt{c^2+2cv+v^2-4v}}\right)^t}{\left(\sqrt{c^2+2cv+v^2-4v}\left(-c-v+\sqrt{c^2+2cv+v^2-4v}\right)\right)}$$

Le résultat est indigeste. Il y a trop de paramètres en jeu (comme souvent en économie ...). Il est préférable de mener une étude classique de l'équation pour repérer les différentes dynamiques qu'elle peut engendrer.

3. Il suffit d'appliquer la théorie mathématique. Une solution de **eqrY** est de la forme  $Y(t) = k r^t$ , et l'équation caractéristique s'en déduit en remplaçant  $Y(t)$ ,  $Y(t-1)$  et  $Y(t-2)$  par leur expression puis en simplifiant.

```
> eqdis:=subs({Y(t)=r^2,Y(t-1)=r,Y(t-2)=1},eqrY);
carac:=collect(lhs(eqdis)-rhs(eqdis),r)=0;#équation
caractéristique
```

$$\begin{aligned} eqdis &:= r^2 = c r + v r - v \\ carac &:= r^2 + (-c - v) r + v = 0 \end{aligned} \quad (5.4)$$


Passons au calcul du discriminant de l'équation caractéristique. On invoque la commande **discrim** dont le premier argument est une expression (et non une équation) et le second désigne l'inconnue. Des manipulations algébriques simples permettent de simplifier ce discriminant.

```
> disc:=discrim(lhs(carac),r);
disc:=algsols(c^2+2*c*v+v^2=x,%);
disc:=subs(x=(c+v)^2,%);
```

$$\begin{aligned} disc &:= c^2 + 2 c v + v^2 - 4 v \\ disc &:= x - 4 v \\ disc &:= (c + v)^2 - 4 v \end{aligned} \quad (5.5)$$

Trois possibilités : le discriminant est strictement positif, et alors l'équation caractéristique a deux racines réelles; le discriminant est nul, auquel cas l'équation caractéristique admet une racine double; le discriminant est strictement négatif et l'équation caractéristique a deux racines complexes conjuguées. En tenant compte du fait que  $0 < c < 1$  et  $v > 0$ , le premier cas correspond ici à la configuration  $c > -v + 2\sqrt{v}$ , le second à  $c = -v + 2\sqrt{v}$  et le troisième à  $c < -v + 2\sqrt{v}$ . Le graphique suivant synthétise ces observations. Les valeurs du coefficient d'accélération  $v$  sont en abscisse; la propension à consommer  $c$  est en ordonnée; la courbe représentative est celle de la fonction  $v \rightarrow c = -v + 2\sqrt{v}$ , dont on ne retient que la partie significative tracée en noir (correspondant à  $0 < c < 1$  et  $0 < v < 4$ ) pour laquelle le discriminant est nul. La surface en jaune donne les couples  $(v, c)$  pour lesquels le discriminant est positif. La surface en rouge donne les couples  $(v, c)$  pour lesquels le discriminant est négatif.

```
> with(plots):
graf1:=implicitplot(disc,v=0..5,c=-0.5..1,view=[0..5,-0.5..1.2],thickness=2):
graf1a:=implicitplot(disc,v=0..4,c=0..1,view=[0..5,-0.5..1.2],thickness=2,filled=true):
graf2:=plot(1,v=0..5,color=black):
graf3:=plot([[4,-0.5],[4,1.2]],style=line,color=black):
graf4:=display(graf1,graf1a,graf2,graf3):graf4;
```



4. On commence par créer deux fonctions-procédures pour chacune des racines de l'équation caractéristique.

```
> rac:=solve(carac,r);
  rac[1];racin1:=unapply(rac[1],(v,c));
  rac[2];racin2:=unapply(rac[2],(v,c));
```

$$rac := \frac{1}{2}c + \frac{1}{2}v + \frac{1}{2}\sqrt{c^2 + 2cv + v^2 - 4v}, \frac{1}{2}c + \frac{1}{2}v - \frac{1}{2}\sqrt{c^2 + 2cv + v^2 - 4v}$$



$$racin1 := (v, c) \rightarrow \frac{1}{2}c + \frac{1}{2}v + \frac{1}{2}\sqrt{c^2 + 2cv + v^2 - 4v}$$

$$racin2 := (v, c) \rightarrow \frac{1}{2}c + \frac{1}{2}v - \frac{1}{2}\sqrt{c^2 + 2cv + v^2 - 4v} \tag{5.6}$$

Si le discriminant est strictement positif, on a deux racines réelles positives car leur produit vaut  $v > 0$  et leur somme vaut  $c + v > 0$ . La trajectoire de  $Y$  est convergente si les deux racines sont inférieures à l'unité, divergente sinon.


```
> plot(racin1(0.25,0.8)^t+racin2(0.25,0.8)^t,t=0..20);#cas de
  la convergence
  plot(racin1(3.5,0.8)^t+racin2(3.5,0.8)^t,t=0..2);#cas de la
```

## divergence




Si le discriminant est nul, la trajectoire de  $Y$  est toujours divergente puisque la solution est de la forme  $k_1\sqrt{v}^t + k_2 t\sqrt{v}^t$ .


```
> plot(racine1(2,2*sqrt(2)-2)^t+t*racine2(2,2*sqrt(2)-2)^t,t=0.  
.3);
```



Enfin, quand le discriminant est négatif, les deux racines sont complexes. Si leur module est inférieur à 1, la trajectoire de  $Y$  converge en oscillant. Sinon, la trajectoire est oscillatoire et divergente.

```
> plot(racin1(0.9,0.8)^t+racin2(0.9,0.8)^t,t=0..100);#cas
oscillatoire convergent
plot(racin1(1.1,0.8)^t+racin2(1.1,0.8)^t,t=0..100);#cas
oscillatoire divergent
```





## Exercice E2 (Evolution du prix d'un titre)

### Enoncé

Pour toute période  $t$ , on note  $P(t)$  le prix d'une action et  $P^a(t+1)$  son prix anticipé pour la période suivante. Ce titre rapporte un dividende constant  $d$  à chaque période.

Un opérateur sur le marché des titres compare le prix de ce titre  $P(t)$  et la somme actualisée au taux constant  $i$  ( $0 < i < 1$ ) du dividende et de sa valeur prévue de revente, soit  $\frac{1}{1+i}(d + P^a(t+1))$ .

L'arbitrage se définit par la condition  $P(t) = \frac{1}{1+i}(d + P^a(t+1))$  puisque la somme investie à la période  $t$  est égale, du point de vue de l'opérateur, à la somme récupérée à la période  $t+1$ .

1. Le prix du titre est à son niveau d'équilibre  $Pe$  si le prix anticipé est égal au prix courant :

$Pe = P(t) = P^a(t+1)$ . Calculer ce prix d'équilibre. Montrer qu'il est égal à la somme infinie actualisée des dividendes que l'action rapporte. Comment interpréter ce résultat?

2. On suppose que les opérateurs du marché font des anticipations adaptatives sur le prix du titre. Le prix anticipé pour la période suivante est égal au prix qu'il avait anticipé pour la période courante corrigé de l'erreur d'anticipation constatée :  $P^a(t+1) = P^a(t) + h(P(t) - P^a(t))$  avec  $0 < h < 1$ .

a) Donner la loi séquentielle d'évolution du prix du titre.

b) Résoudre cette équation de récurrence.

c) Montrer que toutes les trajectoires du prix convergent vers le prix d'équilibre.

d) Donner une représentation graphique du phénomène.

3. On suppose à présent que les opérateurs font des anticipations rationnelles parfaites. Le prix prévu est toujours égal au prix de la période suivante :  $P^a(t+1) = P(t)$ .

a) Donner la loi séquentielle d'évolution du prix du titre.

b) Résoudre cette équation de récurrence.

c) A quelle condition le prix du titre converge vers sa valeur d'équilibre? Comment se produit une

bulle spéculative?

d) Donner une représentation graphique du phénomène.

### Solution

1. On commence par poser l'équation d'arbitrage, qu'on résout avec **solve** sous la condition d'égalité du prix anticipé au prix courant.

```
> restart;  
eq1:=P(t)=1/(1+i)*(Pa(t+1)+d);  
Pe:=solve(subs(Pa(t+1)=P(t),eq1),P(t));
```

$$eq1 := P(t) = \frac{Pa(t+1) + d}{1 + i}$$

$$Pe := \frac{d}{i} \quad (6.1)$$

Manifestement, le prix d'équilibre est égal à la somme infinie des termes d'une suite géométrique de premier terme  $\frac{d}{1+i}$  et de raison  $\frac{1}{1+i}$  :  $Pe = \sum_{k=1}^{\infty} \frac{d}{(1+i)^k}$ . Une fois n'est pas coutume, on charge

le paquetage (riche en possibilités) **SumTools** pour accéder à la commande **Summation** afin de confirmer ce résultat.

```
> with(SumTools):  
Summation(d/(1+i)^k,k=1..infinity);
```

$$\frac{d}{i} \quad (6.2)$$

Ainsi, le prix d'équilibre de l'action est la somme actualisée des dividendes qu'elle rapporte. Il s'interprète comme la valeur fondamentale du titre.

2.

a) Quand les anticipations sont adaptatives, on dispose de deux équations pour déterminer la loi d'évolution séquentielle du prix du titre.

```
> eq2:=subs(t=t-1,eq1);  
eq3:=Pa(t+1)=Pa(t)+h*(P(t)-Pa(t));#formation des  
anticipations  
Pa(t+1):=solve(eq1,Pa(t+1));#assignation de Pa(t+1)  
Pa(t):=solve(eq2,Pa(t));#assignation de Pa(t)  
eqr:=P(t)=solve(eq3,P(t));#équation de récurrence en P(t)
```

$$eq2 := P(t-1) = \frac{Pa(t) + d}{1 + i}$$

$$eq3 := Pa(t+1) = Pa(t) + h (P(t) - Pa(t))$$

$$Pa(t+1) := P(t) + P(t) i - d$$

$$Pa(t) := P(t-1) + P(t-1) i - d$$

$$eqr := P(t) = \frac{-P(t-1) - P(t-1) i + h P(t-1) + h P(t-1) i - h d}{-1 - i + h} \quad (6.3)$$

b) On fait appel à **rsolve** pour résoudre cette équation de récurrence linéaire du premier ordre à coefficients constants et second membre constant.

```
> P(t):=simplify(rsolve(eqr,P(t)));
```



$$P(t) := \frac{P(0) \left( \frac{-1-i+h+hi}{-1-i+h} \right)^t i + d - d \left( \frac{-1-i+h+hi}{-1-i+h} \right)^t}{i} \quad (6.4)$$

Au prix de quelques manipulations, on donne une forme plus agréable/lisible à la solution.

```
> P(t):=subs((-1-i+h+h*i)/(-1-i+h)^t=b,P(t));
P(t):=expand(P(t));
P(t):=collect(P(t),b);
P(t):=subs(b=(-1-i+h+h*i)/(-1-i+h)^t,P(t));
```

$$P(t) := \frac{P(0) b i + d - d b}{i}$$

$$P(t) := P(0) b + \frac{d}{i} - \frac{d b}{i}$$

$$P(t) := \left( P(0) - \frac{d}{i} \right) b + \frac{d}{i}$$

$$P(t) := \left( P(0) - \frac{d}{i} \right) \left( \frac{-1-i+h+hi}{-1-i+h} \right)^t + \frac{d}{i} \quad (6.5)$$

On en tire la suite-solution comme fonction-procédure à l'aide de **unapply**.

```
> p:=unapply(P(t),t);
```

$$p := t \rightarrow \left( P(0) - \frac{d}{i} \right) \left( \frac{-1-i+h+hi}{-1-i+h} \right)^t + \frac{d}{i} \quad (6.6)$$

c) Il est temps d'introduire les hypothèses sur les paramètres  $i$  et  $h$ .

```
> assume(0<i,i<1,0<h,h<1);
```

Des tests directs montrent que  $\frac{-1-i+h+hi}{-1-i+h}$  est toujours un nombre de l'intervalle ouvert ]0; 1[.

```
> is((-1-i+h+h*i)/(-1-i+h)<1);
is((-1-i+h+h*i)/(-1-i+h)>0);
```

*true*

*true*

(6.7)

En conséquence, le prix de l'actif converge vers son niveau d'équilibre  $\frac{d}{i}$  pour tout  $P(0)$ . Quand les anticipations sont adaptatives, la valeur fondamentale de l'action correspond à un équilibre dynamique globalement stable.

d) A partir de l'expression de la solution générale, on crée avec **unapply** une fonction-procédure à deux variables :  $t$  et  $P(0)$ . Comme Maple refuse un argument du type  $P(0)$ , on renomme le prix de départ par  $P0$ .

```
> P(t):=subs(P(0)=P0,P(t));
p:=unapply(P(t),(t,P0));
```

$$P(t) := \left( P0 - \frac{d}{i} \right) \left( \frac{-1-i+h+hi}{-1-i+h} \right)^t + \frac{d}{i}$$


(6.8)

$$p := (t, P0) \rightarrow \left( P0 - \frac{d}{i} \right) \left( \frac{-1 - i + h + h i}{-1 - i + h} \right)^t + \frac{d}{i} \quad (6.8)$$

On pourra alors tracer facilement plusieurs trajectoires dans le même graphe par variation de la condition initiale. Naturellement, il faut préalablement attribuer - assez arbitrairement - des valeurs aux paramètres  $d$ ,  $i$  et  $h$ .

```
> d:=50;#montant du dividende
i:=0.05;#taux d'actualisation
h:=0.15;#paramètre de correction d'erreur d'anticipation
plot({[seq([t,p(t,500)],t=0..500,10)],[seq([t,p(t,750)],t=0.
.500,10)],[seq([t,p(t,1000)],t=0..500,10)],[seq([t,p(t,1250)
],t=0..500,10)],[seq([t,p(t,1500)],t=0..500,10)]},style=
point);
```

```
d := 50
i := 0.05
h := 0.15
```



3. On prépare cette question en désassignant les variables et paramètres et en introduisant l'hypothèse des anticipations rationnelles parfaites.

```
> d,i,h,Pa(t+1),P(t):='d','i','h','Pa(t+1)','P(t)';
#désassignations
eq3:=Pa(t+1)=P(t+1);#formation d'anticipations rationnelles
parfaites
```

```
d, i, h, Pa(t + 1), P(t) := d, i, h, Pa(t + 1), P(t)
eq3 := Pa(t + 1) = P(t + 1)
```

(6.9)

a) La loi d'évolution du prix du titre est l'équation de récurrence obtenue en injectant l'expression du prix anticipé déduite de l'équation **eq1** dans l'équation des anticipations **eq3**.

```
> Pa(t+1):=solve(eq1,Pa(t+1));
eqr:=eq3;#loi d'évolution séquentielle du prix de l'action
Pa(t+1) := P(t) + P(t) i - d
eqr := P(t) + P(t) i - d = P(t+1)
```

$$(6.10)$$

b) La résolution de **eqr** ne présente aucune difficulté pour **dsolve**.

```
> P(t):=rsolve(eqr,P(t));
```

$$P(t) := P(0) (1+i)^t - \frac{d(1+i)^t}{i} + \frac{d}{i}$$

$$(6.11)$$

On arrange l'allure de la solution par quelques manipulations.

```
> P(t):=subs((1+i)^t=b,P(t));
P(t):=collect(P(t),b);
P(t):=subs(b=(1+i)^t,P(t));
```

$$P(t) := \left( P(0) - \frac{d}{i} \right) b + \frac{d}{i}$$

$$P(t) := \left( P(0) - \frac{d}{i} \right) b + \frac{d}{i}$$

$$P(t) := \left( P(0) - \frac{d}{i} \right) (1+i)^t + \frac{d}{i}$$

$$(6.12)$$

c) Bien que ce soit évident, on demande à Maple de confirmer que le processus est explosif.

```
> assume(i>0,i<1);
is(1+i>1);
limit((1+i)^t,t=infinity);
```

$$\begin{matrix} true \\ \infty \end{matrix}$$

$$(6.13)$$

En fait, trois cas se présentent : si  $P(0) = \frac{d}{i}$ , la trajectoire  $P(t)$  est une suite constamment égale à la valeur fondamentale; si  $P(0) > \frac{d}{i}$ , la trajectoire  $P(t)$  est une suite monotone croissante qui tend vers l'infini et s'interprète comme une bulle spéculative; si  $P(0) < \frac{d}{i}$ , la trajectoire  $P(t)$  est une suite monotone décroissante qui tend vers  $-\infty$ .

d) La représentation graphique de ces trois cas est aisée. On commence par créer une fonction-procédure de deux variables indiquant que la suite-solution dépend du temps et de la condition initiale.

```
> P(t):=subs(P(0)=P0,P(t));#remplacement obligatoire de P(0)
par P0
p:=unapply(P(t),(t,P0));#création d'une fonction-procédure
```

$$P(t) := \left( P0 - \frac{d}{i} \right) (1+i)^t + \frac{d}{i}$$

$$p := (t, P0) \rightarrow \left( P0 - \frac{d}{i} \right) (1+i)^t + \frac{d}{i}$$

$$(6.14)$$

Après avoir calé les paramètres, on invoque `plot`.

```
> d:=50;#montant du dividende  
i:=0.05;#taux d'actualisation  
plot({[seq([t,p(t,500)],t=0..50,2)], [seq([t,p(t,750)],t=0.  
.50,2)], [seq([t,p(t,1000)],t=0..50,2)], [seq([t,p(t,1250)],t=  
0..50,2)], [seq([t,p(t,1500)],t=0..50,2)]}, style=point);  
d:=50  
i:=0.05
```

